SysOp Tools Learning Guide: 

Fix User Accounts Set as System Accounts - Changing the UserAccountControl AD Attribute via script from 544 to 512
Problem: User Accounts in Active Directory show up as System Accounts because they are set with the Password_Not_Required attribute on userAccountControl
An issue we see regularly with domains that contain user accounts which were migrated from NT4 or Novell using 3rd-party migrations tools is the user objects show incorrectly as “SystemAccount” with an added attribute of “Password_Not_Required”. When the accounts are migrated into AD, the userAccountControl attribute is not properly set to 512 (an active normal user account object that requires a password) or 514 (a disabled normal user account object), causing the issue. We also see this happen frequently with home-grown user account creation scripts or other software that does not set the user account attribute value on creation. Essentially, if the userAccountControl attribute is not created for the object at time of object creation in AD, Active Directory automatically  assigns the account an attribute value of 544 and classifies it as a ‘System’ object / Password Not Required. Not good.
 
What are the ramifications? A malicious user could figure out that a password is not required on these accounts, and script them all to be set with a blank password. Nor further explanation is needed with just how bad this could be for domain security. As further issue examples, our password expiration reminder software Password Reminder PRO will not send expiring password reminders to System accounts, because system accounts are not supposed to belong to a human user. Likewise, our Password Reset PRO software will not allow self service access to System accounts. More than likely, many other user account management software products will not categorizing these accounts properly either.

What is userAccountControl and why is it important?

The actual attribute field for AD user accounts we're discussing here is called userAccountControl.
This field identifies whether the user object should be treated by AD as a non-user "system" account or treated as a normal user account. A "system" account tells AD that a password is not required by the account (even though it may have a password set), and AD assumes the password is managed by the system process which created that account. These system accounts are typically created, managed and used by an integrated root service, domain process or inter-domain process (IUSR, IWAM, for example).

The userAccountControl field value for a normal domain user account is 512 (password is required), but the accounts in question may be set to 544 which is 512 plus 32 (a password is not required).
This presents a potential security risk by allowing a user to maintain a domain logon account with a blank password, because, AD is instructed that a password is not required on this account.

To fix the issue of normal user accounts having an incorrect attribute set, you will have to go through the schema with ADSIEDIT and change these user objects' userAccountControl value to 512, and the problem will be resolved. Changing this field value will not affect anything that your users will notice. However, using ADSIEDIT is dangerous and should be used with care as you are basically editing the ‘registry’ of your domain! Additionally, doing this one-by-one can take a long time if you have many incorrectly flagged accounts. With 2008 AD, the ADUC console now has an “Attribute Editor” tab, allowing you direct access to fix the userAccountControl value on each user object a bit quicker.

How about fixing thousands of incorrect user objects? Fear not- Below you will find a VB Script that you can easily modify, point to an OU of user objects, automatically discover “544” users, and force-set the correct attribute of 512.  This is much easier!  But do be careful- you DO NOT want to run this script on your entire domain, or you may break any real ‘system’ accounts that need to be there. ONLY run it on an OU that you know contains problem user objects, or on the entire domain if you are CERTAIN there are no legitimate system accounts with a 544 value. Most of the time, only your messed up user accounts will have “544” for a userAccountControl value, standard system accounts will have different userAccountControl values.

 

Let’s change things!
 

Prerequisites for changing the userAccountControl attribute from 544 to 512
Logon to a DC or member server as a domain administrator, edit the script to suit your domain, and run the script from an elevated command prompt. 
Instructions for using this script in a 2003, 2008, 2008R2 domain
1.   Copy and paste the example script below into notepad. 

2.   Change the value for “OU=My 544 Users,DC=mydomain,DC=local “ to point to the OU of your problem user objects. You must leave a comma and space after the OU name and before the end quote, and correct CapiTaLization of OU names. Always start OU= with the downlevel OU first (backwards from how it is laid out up to top of tree in your AD Users and Computers). You can remove the OU path and just use “DC=mydomain,DC=local”, but it is always preferred to control the script by pointing to an OU that only contains known human user accounts.
3.   Change the log output path in the script to your preferred destination. The default is C:\
4.   Save the file with a .vbs extension, for example: Fix544.vbs. 

5.   Open an elevated CMD prompt preferably on the DC, change path to script, e.g. “C:\Fix544.vbs” and hit enter. Script will present a dialogue indicating it is running. Click OK to continue. A final prompt will present when finished. Read the data in the output csv log file.
Script is on next page…

Sample Script to change userAccountControl from 544 to 512 (copy area in blue and paste in notepad)

 START COPY BELOW THIS LINE 
' FixUserAccountControl.vbs  (use cscript to execute this from an elevated cmd prompt preferably on the DC)

'

' This script will search the through the domain or specific OU path to process every User marked with 544.  

‘ For each user with UserAccountControl = 544 it will reset it to 512 and log the activity in a text file for later review. 

‘ This change will be transparent to the end user

' An immediate notice is written to screen as it searches to let you know it is still working.

' When finished, A final line is written to screen with the count of how many users were changed.

' MODIFY THE DOMAIN NAME AND LOG DIRECTORY PATH AS NOTED BELOW - YOU CAN SPECIFY AN OU PATH IF NEEDED

' SCRIPT PROVIDED AS-IS AND WITHOUT WARRANTY - YOU ASSUME ALL RISK RUNNING THIS IN YOUR DOMAIN SO BE CAREFUL

Option Explicit

Const ADS_SCOPE_SUBTREE = 2

Const ForWriting = 2

Dim outfile

Dim oConn, oCmd

Dim name

Dim oFSO, oFile

Dim oRS, oRS2

Dim query

Dim outputStr

Dim accountControl, status

Dim strDN, DN

Dim oUser

Dim FixCount

strDN   = "OU=PWD Expiring Users,DC=sysoptools,DC=net"   ''' CHANGE TO MATCH YOUR DOMAIN AND OU PATH

outfile = "C:\544FixReport.csv"                          ''' CHANGE IF NEEDED FOR LOG OUTPUT DESTINATION

Set oFSO  = CreateObject("Scripting.FileSystemObject")

Set oFile = oFSO.OpenTextFile(outfile, ForWriting, True)

oFile.WriteLine Chr(34) & "dn" & Chr(34) & "," & _

                Chr(34) & "username" & Chr(34) & "," & _

                Chr(34) & "status" & Chr(34)

Set oConn = CreateObject("ADODB.Connection")

Set oCmd = CreateObject("ADODB.Command")

oConn.Provider = "ADsDSOObject"

oConn.Open "Active Directory Provider"

Set oCmd.ActiveConnection = oConn

oCmd.Properties("Page Size") = 1000

oCmd.Properties("Searchscope") = ADS_SCOPE_SUBTREE

oCmd.CommandText = "SELECT distinguishedName FROM 'LDAP://" & strDN & "' WHERE objectCategory = 'organizationalunit'"

Set oRS = oCmd.Execute

FixCount = 0

oRS.MoveFirst

Do Until oRS.EOF

   DN = oRS.Fields("distinguishedName").Value

   CheckUserAccountControl DN

   oRS.MoveNext

Loop

oFile.Close

Wscript.Echo "Now running - Click OK to continue, you'll see another prompt when completed. "

Wscript.Echo "Finished!  Number of accounts changed from 544 to 512 = " & FixCount

Sub CheckUserAccountControl(baseDN)

query = "<LDAP://" & baseDN & _

        ">;(&(ObjectClass=User)(userAccountControl=*));" & _

        "distinguishedName,samAccountName,userAccountControl;subtree"

Set oRS2 = oConn.Execute(query)

While Not oRS2.Eof

   ' Get the value for userAccountControl, convert to a

   ' Long because the value of userAccountControl
   ' might be greater than the maximum value for an Int.

   accountControl = CLng(oRS2("userAccountControl"))

  ' outputStr = Chr(34) & oRS2("distinguishedName") & Chr(34) & "," & _

  '             Chr(34) & oRS2("samAccountName") & Chr(34) & "," & _

  '             Chr(34) & " was " & accountControl & Chr(34) & ","

  ' Wscript.Echo outputStr                           ''' I'm still working...

   If accountControl = 544 Then

      Set oUser = GetObject("LDAP://" & oRS2("distinguishedname"))

      oUser.Put "userAccountControl", 512

      oUser.SetInfo

   '   outputStr = outputStr & " changed to 512"

   '   Wscript.Echo "     *****  " & outputStr       ''' eyecatcher

      oFile.WriteLine outputStr

      FixCount = FixCount + 1

   End If

   oRS2.MoveNext

Wend

End Sub 
END COPY ABOVE THIS LINE – 

!! Be sure after you have pasted into notepad that all line breaks are same as above !!
Notes on Making Changes With The Above Script 
1.   Purely for testing, we suggest setting the script to point to a new OU called TEST with one user account in it that has a “544” value.  Then open up Active Directory Users and Computers after running the script to ensure the proper change was made and the value now shows 512. 
 

2.   If you are having a tough time finding all of your incorrectly-attributed user accounts in AD, and would like an easy method to globally-view and validate successful changes made by the script, please feel free to use our Password Reminder PRO or Password Reset PRO software as these products contain a user account audit console that will display (1) all “system” user accounts with the incorrect attribute and (2) show them as normal fixed accounts if the script was successful. Our software is FREE to use for 30 days, more than enough time for you to plan and execute domain-wide fixes, no catch or purchase involved. 

 3.  Before running the script, run Password Reminder PRO's User Report Console and notice that all of the incorrectly attributed user objects will show under the 'Misc Accounts' tab view, with the box on far right under the column of 'SystemAccount' checked. These accounts will be the focus of your fix as the check mark indicates the account has an attribute of 544 (system account).  
 

 4.  After you run the above script on these accounts and set the attribute to 512, you will see them disappear from the Misc Accounts view and appear under 'Licensed Users' view. Your users will *require* a password, and domain security / harmony is preserved. Success!
Reference:  UserAccountControl needs a numeric value in order to set the account cataloguing properly within AD, and enforce certain security mechanisms. For example, two common values for userAccountControl are: 512 = enabled password expiring account and 514 = disabled user account. Read the MS Article on all possible attribute values for userAccountControl

